

Save and Backup Files with a Macro
WordPerfect Magazine
By S. Scott Zimmerman

The included ALTS.WPM macro makes saving and backing up your documents hassle free.

Using the macro

As you're preparing a document with WordPerfect, just press (Alt-S) from time to time. If you've never saved the file before, pressing (Alt-S) has the same effect as pressing Save (F10). WordPerfect displays the prompt Document to be saved:, after which you type the name of your document and press (Enter).

If you've already saved the file (so it has a filename displayed in the lower left corner), pressing (Alt-S) automatically saves the updated document on-screen and replaces the outdated document on the disk.

As you can see, pressing (Alt-S) instead of Save (F10), (Enter), (Y) Yes saves two keystrokes. This is no big deal, except for us paranoids who like to save our screen documents every few minutes. Over the course of a day or a week, the time saved adds up.

Now for the best part. If you press (Alt-S) and the screen document is up-to-date – that is, you haven't modified the screen document since the last time you saved it – the macro displays the prompt Doc is up-to-date. 1 Copy to Backups Directory; 2 Copy to Drive A: 0.

Now you can press any key such as 0 or (Enter) and no save will take place. Or, you can type 1 or A to copy the document to the "c:\wp51\backups" directory or type 2 or B to copy to the root directory of drive A.

Good news and bad news

The good news about the macro is its efficiency in saving and backing up documents you're still working on. With a few keystrokes you can save the document on-screen to a hard disk and save a copy of it on a floppy disk. You'll find this especially handy when you have to transfer a document file from your computer at work to your one at home and back again.

The bad news is that the macro never asks you to verify a replacement. If the macro detects a disk file with the same name as the screen document, the macro simply overwrites the disk file.

The only exception is the first time you save the document. If there's already a document with the same name at that time, WordPerfect will still warn you. Actually, you should experience no difficulty if you're careful to give document files unique names.

Understanding the macro

The macro checks several things before deciding which action to take. Line 1 checks to see if there's an error during macro execution. If an error occurs, such as a disk drive door open when you try to back up a file, then you have to respond to the DOS error message 1 Retry; 2 Cancel, and the macro goes to {LABEL}Error (line 40). This subroutine ensures that the macro quits at the normal editing screen. Lines 2 through 5 check that you're at a normal editing screen when you start the macro.

If you're at some other screen, such as the Format menu, when you run the macro, line 4 sounds a bell and the macro quits.

Next, the macro checks to see if there's anything on the screen (lines 6-10). If the screen is blank, there's nothing to save, so the macro sounds a bell and quits.

Line 11 then checks if the macro has a name. If not, line 12 executes the standard Save (F10) command and quits. You can then type a name and press (Enter) to save the document.

Line 14 is where things get a bit more complex. This line assigns to the variable "mod" the value of a system variable that checks to see if the document has been modified since the last time it was saved, and then line 15 assures that the save wasn't simply an automatic backup.

If the document has not been properly saved, the macro then executes the Save (F10) command, accepts the file's original name and replaces the older version.

If the document hasn't been modified since the last save, the {ELSE} command on line 18 takes over. The commands on lines 19 through 22 display the prompt Doc is up-to-date. 1 Copy to drive A; 2 Copy to drive B: O. The {CASE} command on lines 23 and 24 branches the macro depending upon which key you pressed in answer to the prompt. If you pressed "1," the macro continues on line 27 with {LABEL}CopyA. If you pressed "2," the macro continues on line 30 with {LABEL}CopyB.

If you pressed any other key, line 25 ends the macro without performing any save.

Each of the subroutines on lines 27-29 and 30-32 assigns a different value to the variable "place" and then passes control to the next routine on line 33 labeled "CopyIt." This subroutine copies the updated file to the desired location using List (F5). Lines 35 and 36 list the current file's directory and search for that file. Line 37 then copies it to the place specified by the variable "place."

Finally, line 38 uses the {STATE} command to find out if it's in a Yes/No menu such as Replace filename? No (Yes). If so, it replaces the file by answering (Y) Yes.

The final subroutine (lines 40-43) handles any errors, as explained above.

Customizing the macro

If you want to change the directories or drive letters to which your documents get backed up, the highlighted lines below (inside the red box) will need to be changed.

To do so, go into the Macro Editor – press (Ctrl-F10), type "alts" and press 2 – then make the appropriate changes. Press F7 to save and exit the macro.

Note: The line numbers to the left of the red box are only for identification purposes; you will not see them in the Macro Editor.

```

1 {ON ERROR}{GO}Error~~
2 {IF}{STATE}&4~
3 {ELSE}
4     {BELL}{QUIT}
5 {END IF}
6 {IF}{SYSTEM}Document~&256~
7     {BELL}{PROMPT}Document screen is blank. ~
8     {WAIT}35~
9     {QUIT}
10 {END IF}
11 {IF}"{SYSTEM}Name~"=""~
12     {Save}{QUIT}
13 {END IF}
14 {ASSIGN}mod~{SYSTEM}Document~&3~
15 {IF}{VARIABLE}mod~!=0~
16     {Save}{Enter}Y
17     {QUIT}
18 {ELSE}
19     {CHAR}back~Document is up to date.
20     1 Copy to Backups directory:
21     2 Copy to Drive A
22     0{Left}{BELL}~
23     {CASE}{VARIABLE}back~~
24     1~Backups~2~DriveA~~
25     {QUIT}
26 {END IF}

27 {LABEL}Backups~
28     {ASSIGN}place~c:\wp51\backups~
29     {GO}CopyIt~
30
31 {LABEL}DriveA~
32     {ASSIGN}place~a:\~
33     {GO}CopyIt~
34
35 {LABEL}CopyIt~
36     {DISPLAY OFF}
37     {List}{SYSTEM}path~{Enter}
38     n{SYSTEM}name~{Enter}
39     8{VARIABLE}place~{Enter}
40     {IF}{STATE}&1024~y{END IF}
41     {Exit}{QUIT}
42
43 {LABEL}Error~
44     {IF}{STATE}&4~{QUIT}
45     {ELSE}{Exit}{GO}Error~
46 {END IF}

```